

Nastavni predmet:	Ugradbeni računalni sustavi
Vježba: 12	Inkrementalni enkoder
Cilj vježbe:	Upravljanje vanjskim elementima spojenih na Arduino pomoću inkrementalnog enkodera

Upute

Sve zadatke spremi na USB, a u bilježnici za sve zadatke napiši:



- postupak izrade programa
- objašnjenje korištenih naredbi
- dobivene rezultate po točkama
- odgovoriti u bilježnicu na postavljena pitanja vezana uz ovu vježbu
- Ukoliko u kòdu postoji greška, korigiraj i objasni!

Inkrementalni okretni davači (enkoderi) su senzori koji mogu detektirati kutnu poziciju i brzinu pretvarajući rotacijski mehanički pomak u električne impulse. Moguće je detektirati i **smjer okretanja** uspoređujući dva impulsa koji su fazno pomaknuti za 90°.

Za razliku od potencijometra, imaju **neograničen broj okretaja u oba smjera**. Često se isporučuju s osovnom kao tipkalom koje se može koristiti za potvrdu odabira izbornika, kao reset tipka ili kao bilo koja tipka prikladna u danom trenutku.


Kad se disk okreće pinovi A i B rade kontakt sa zajedničkim pinom, kao dva tipkala ili dvije sklopke.

U skladu s tim na izlazu A i B će se pojavljivati pravokutni impulsi.




Električna shema KY-040 enkodera

Ako se osovina okreće u smjeru kazaljke na satu (**CW**), impuls na izlazu A će prethoditi izlazu B (za 90° ili T/4). Padajući ili rastući brid impulsa A uvijek prelazi u suprotno stanje nego što je impuls B ($A = \neg B$)




Ako se osovina okreće suprotno od smjera kazaljke na satu (**CCW**), impuls na izlazu B će prethoditi izlazu A (za 90° ili T/4). Padajući ili rastući brid impulsa A uvijek prelazi u isto stanje kao što je i impuls B ($A = B$)



Fotografije: www.howtomechatronics.com


Zadatak 1. Spoji inkrementalni enkoder prema priloženoj shemi i promatraj redoslijed paljenja LED dioda ovisno o smjeru vrtnje. Osovinu okretati sporo i pratiti paljenje LED dioda između dva utora. U ovom zadatku nema koda.

Električna shema



Zadatak 2. Proširi spoj iz prethodnog zadatka povezivanjem inkrementalnog enkodera s mikroupravljačem. Stanje pinova CLK, DT prikazati na Serial monitoru.

Električna shema



Kod zadatka

```
const int PinCLK = 2; // Izlaz A enkodera - CLK signal
const int PinDT = 3; // Izlaz B enkodera - DT signal
const int PinSW = 8; // Tipkalo enkodera
```

```
int counter = 0;
int iCurrentState;
int iLastState;
```

```
String sPrethodni;
```

```

void setup() {
    pinMode(PinCLK, INPUT);
    pinMode(PinDT, INPUT);

    Serial.begin(9600);
    Serial.println("Ready...");
    // očitava inicijalno stanje izlaza A enkodera
    // izvan loop funkcije
    iLastState = digitalRead(PinCLK);
}

void loop() {
    iCurrentState = digitalRead(PinCLK); // očitava trenutno stanje izlaza
                                         // A enkodera
    // ako se prethodno stanje i trenutno stanje izlaza A razlikuje, znači da
    // imamo impuls
    if (iCurrentState != iLastState) {

        // ako je izlaz B enkodera različit od izlaza A,
        // znači da je smjer okretanja CW
        if (digitalRead(PinDT) != iCurrentState) {
            counter++;
            sPrethodni = "++";
        }
        else {
            counter--;
            sPrethodni = "--";
        }
        Serial.print("\nPinCLK = ");
        Serial.print(digitalRead(PinCLK));
        Serial.print("\tPinDT = ");
        Serial.print(digitalRead(PinDT));
        Serial.print("\tPosition: ");
        Serial.print(counter);
        Serial.print("\tCounter");
        Serial.print(sPrethodni);
    }

    iLastState = iCurrentState; // sadašnje stanje varijable iCurrentState
                                // dodjeljujemo varijabli iLastState i
                                // time smo spremni za novo očitavanje
}

```

Zadatak 3. Program iz prethodnog zadatka modificiraj tako da koristiš prekid INT0 koji se nalazi kao sekundarna funkcija na pinu 2. Ispisivati na Serial monitor. Odgovara li broj impulsa broju koraka enkodera? Objasni!

Cilj je uočiti da ako nema debouncinga, broji više impulsa nego što to stvarno jest!

1. Koja je razlika ako se pri najavi prekida umjesto RISING koristi CHANGE?

Kòd zadatka

```

const int PinCLK = 2;           // Koristimo za generiranje prekida
                                // preko CLK signala (pin A enkodera)

```

```
const int PinDT = 3;          // Koristimo za očitavanje DT signala (pin B
                              // enkodera)
const int PinSW = 8;          // Koristi se za tipkalo

volatile long      virtualPosition = 0;    // obavezno deklarirati kao
                                          // volatile kad se koristi u isr

void isr0() {
    detachInterrupt(0);
    if (!digitalRead(PinDT))
        virtualPosition++;
    else
        virtualPosition--;
    attachInterrupt(0, isr0, RISING);
} // ISR0

void setup() {
    pinMode(PinCLK, INPUT);
    pinMode(PinDT, INPUT);
    pinMode(PinSW, INPUT_PULLUP);
    attachInterrupt(0, isr0, RISING);
    // Deklariramo prekidnu rutinu ISR isr0 koja se
    // izvršava pojavom rastućeg brida na INT 0
    // Testirati ponašanje sklopa ako se umjesto RISING koristi CHANGE
    // Na Arduino UNO interrupt 0 je uvijek
    // spojen na pin 2
    Serial.begin(9600);
    Serial.println("Start");
}

void loop() {
    int lastCount = 0;
    while (true) {
        if (!(digitalRead(PinSW))) {      // provjera da li je stisnuto
                                          // tipkalo
            virtualPosition = 0;          // ako da, resetiraj
                                          // brojač na 0
            while (!digitalRead(PinSW)) { // čekaj dok tipkalo
                                          // nije otpušteno
            }

            Serial.println("Reset");
        }

        if (virtualPosition != lastCount) {
            lastCount = virtualPosition;
            Serial.print("Count:");
            Serial.println(virtualPosition);
        }
    } // while
}
```

Zadatak 4. Modificiraj `isr0()` iz prethodnog zadatka tako da inkrementalni enkoder broji impulse u skladu sa stvarnim zakretom osovine enkodera. Funkcija `isr0()` modificirana je da radi softverski debounce. Pratiti sada na Serial monitoru.


Modificirani kòd funkcije `isr0()`

```
void isr0() {

    detachInterrupt(0);
    static unsigned long lastInterruptTime = 0;
    unsigned long interruptTime = millis();
    // Ako se INT 0 pojavi unutar 5ms, pretpostavlja se da je
    // istitravanje i ignororamo ga
    if (interruptTime - lastInterruptTime > 5) {
        if (!digitalRead(PinDT))
            virtualPosition++;
        else
            virtualPosition--;
    }

    lastInterruptTime = interruptTime;
    attachInterrupt(0, isr0, RISING);
} // ISR0
```

Zadatak 5. Modificiraj spoj i program tako da pomoću inkrementalnog enkodera mijenjaš svjetlinu dviju dioda. Svjetlina dioda treba biti suprotna. Kad jedna svijetli najjače, druga svijetli najslabije. Svjetlinu podešavati pomoću PWM. Vrijednost svjetline jedne diode ispisuj na LCD display.



Zadatak 7. Modificiraj spoj tako da u EEPROM upišeš zadnju vrijednost svjetline diode prije isključenja mikroupravljača. Prilikom uključenja Arduina program treba učitati pohranjenu vrijednost svjetline i omogućiti daljnju promjenu svjetline pomoću rotary enkodera. Koristiti EEPROM.h biblioteku.

Zadatak 8. Modificiraj spoj i program tako da pomoću inkrementalnog enkodera mijenjaš boje RGB diode, RGB vrijednosti ispisuješ na LCD i pohranjuješ vrijednosti u EEPROM. Podešavanje pojedine boje prebacuješ pritiskom na tipkalo inkrementalnog enkodera. Pritiskom dužim od 2 sekunde resetiraš diodu da svijetli bijelom bojom.

Zadatak 9. Modificiraj program tako da pomoću inkrementalnog enkodera i LCD-a napraviš izbornik pomoću kojeg ćeš podešavati

- a) brzinu treptanja RGB diode u koracima 100, 200, 500 i 1000ms,
- b) mijenjati intenzitet crvene boje,
- c) zelene boje i
- d) plave boje.

Promijenjene vrijednosti treba primijeniti odmah po promjeni.

Za ulazak u podizbornik koristiti tipkalo na inkrementalnom enkoderu. Pritiskom duljim od 2 sekunde vraćaš se u glavni izbornik.